

```
In[27]:= Clear["Global`*"]
```

```
In[30]:= gramschmidt[w_, var_, interval_, n_] :=
(* Density function, variable, interval and number of terms *)
Module[{a, gram, ψ, ϕ},
a[i_, j_] :=
Integrate[w * var^(i + j), Join[{var}, interval], GenerateConditions → False];
gram[m_] := Det[Table[a[i, j], {i, 0, m}, {j, 0, m}]]; gram[-1] = 1;
ψ[m_] :=
Det[Append[Table[a[i, j], {i, 0, m - 1}, {j, 0, m}], var^Range[0, m]] // Simplify];
ϕ[m_] := ψ[m]/Sqrt[gram[m - 1] * gram[m]]; Table[{k, ϕ[k]}, {k, 0, n}]
```

```
In[25]:= gramschmidt[Exp[-x^2], x, {-∞, ∞}, 6] // TableForm
```

```
Out[25]//TableForm=
```

$$\begin{array}{l}
0 \quad \frac{1}{\pi^{1/4}} \\
1 \quad \frac{\sqrt{2} x}{\pi^{1/4}} \\
2 \quad \frac{-1+2 x^2}{\sqrt{2} \pi^{1/4}} \\
3 \quad \frac{x(-3+2 x^2)}{\sqrt{3} \pi^{1/4}} \\
4 \quad \frac{3-12 x^2+4 x^4}{2 \sqrt{6} \pi^{1/4}} \\
5 \quad \frac{x(15-20 x^2+4 x^4)}{2 \sqrt{15} \pi^{1/4}} \\
6 \quad \frac{-15+90 x^2-60 x^4+8 x^6}{12 \sqrt{5} \pi^{1/4}}
\end{array}$$

```
In[26]:= gramschmidt[1, x, {-1, 1}, 3] (* Legendre *) // TableForm
```

```
Out[26]//TableForm=
```

$$\begin{array}{l}
0 \quad \frac{1}{\sqrt{2}} \\
1 \quad \sqrt{\frac{3}{2}} x \\
2 \quad \frac{1}{2} \sqrt{\frac{5}{2}} (-1+3 x^2) \\
3 \quad \frac{1}{2} \sqrt{\frac{7}{2}} x (-3+5 x^2)
\end{array}$$

```
In[32]:= gramschmidt[1/Sqrt[1-x^2], x, {-1, 1}, 3] (* Chebyshev I *) // TableForm
```

```
Out[32]//TableForm=
```

$$\begin{array}{l}
0 \quad \frac{1}{\sqrt{\pi}} \\
1 \quad \sqrt{\frac{2}{\pi}} x \\
2 \quad \sqrt{\frac{2}{\pi}} (-1+2 x^2) \\
3 \quad \sqrt{\frac{2}{\pi}} x (-3+4 x^2)
\end{array}$$

```
In[33]:= gramschmidt[Sqrt[1 - x^2], x, {-1, 1}, 3] (* Chebyshev II *) // TableForm
```

```
Out[33]//TableForm=
```

$$\begin{array}{lcl} 0 & & \sqrt{\frac{2}{\pi}} \\ 1 & 2 & \sqrt{\frac{2}{\pi}} x \\ 2 & & \sqrt{\frac{2}{\pi}} (-1 + 4 x^2) \\ 3 & 4 & \sqrt{\frac{2}{\pi}} x (-1 + 2 x^2) \end{array}$$

```
In[34]:= gramschmidt[Exp[-x], x, {0, ∞}, 3] (* Laguerre *) // TableForm
```

```
Out[34]//TableForm=
```

$$\begin{array}{lcl} 0 & 1 & \\ 1 & -1 + x & \\ 2 & \frac{1}{2} (2 - 4 x + x^2) & \\ 3 & \frac{1}{6} (-6 + 18 x - 9 x^2 + x^3) & \end{array}$$

```
In[35]:= (* gramschmidt[x Exp[-x], x, {0, ∞}, 3] (* Arfken 10.3.4 - 6th E. *) // TableForm *)
```

```
In[36]:=
```